

## ADI-12

Stand: 28.06.94

Dieses Handbuch soll die Funktionsweise und Anwendung der Baugruppe ADI-12 erklären. Text, Schaltung und Platinenlayout sind urheberrechtlich geschützt und dürfen, auch auszugsweise, nur mit ausdrücklicher Zustimmung des Verfassers vervielfältigt oder kopiert werden. Jegliche Haftung für Schäden, durch die Verwendung der Baugruppe, der Software oder des Handbuchs ist ausgeschlossen. Irrtum und technische Änderungen bleiben vorbehalten.

Anregungen und Verbesserungsvorschläge sind jederzeit willkommen!

**Dipl.-Ing. Michael Schmidt**

analoge und digitale Elektronik

---

Aureliusstr.. 22  
52072 Aachen

Tel.: 02 41/ 2 05 22  
Fax.: 02 41/ 40 89 58

.- **Inhaltsverzeichnis** -----  

Funktionsbeschreibung.....1 ff  
Adreßbelegung.....6  
Demoprogramm.....7 ff  
Fehlergrenzen und Meßgenauigkeit.....13 f  
Steckverbindungen.....15 f  
Jumperfunktionen.....17  
GAL-Dokumentation.....18  
Bauteileliste.....19  
Literaturliste.....20

Anhang:

Bestückungszeichnung

Schaltplan Teil 1

Schaltplan Teil 2

Datenblätter

.. Funktionsbeschreibung ADI-12 -----  
|  
-----

Analogteil:

Der zentrale Baustein der Baugruppe ADI-12 ist ein Analog-Digital-Umsetzer vom Typ ICL 7109. Dieser schon etwas betagte Baustein wird u.a. von den Firmen Intersil, Maxim und Harris geliefert und zeichnet sich durch niedrigen Preis, gute Verfügbarkeit und eine sichere und exakte Funktion aus. Der Umsetzer arbeitet nach dem Dual-Slope- oder Zweirampen-Prinzip. Das Verfahren ist relativ langsam in der Umsetzung, läßt aber eine hohe Auflösung zu und ist gleichzeitig sehr genau. Dem Meßsignal überlagerte Rauschstörungen und netz-synchrone Brummspannungen werden bei der Umsetzung unterdrückt, so daß keine Mittelwertbildung über mehrere Meßwerte nötig ist. Das Verfahren wird in fast allen Digitalmultimetern angewandt und eignet sich für die Erfassung von langsam veränderlichen Größen. Insbesondere die Signale von Dehnungsmeßstreifen, Druck-, Feuchte- und Temperatursensoren, Durchflußmessern und potentiometrischen Wegaufnehmern können gut und mit geringem Softwareaufwand bewertet werden.

Details:

Um mehrere Eingangssignale mit einer Baugruppe ADI-12 auszuwerten, wurde dem AD-Umsetzer ein Multiplexer vorgeschaltet. Damit können nacheinander maximal acht verschiedene Eingänge abgetastet werden. Die Eingangssignale passieren zunächst die 100k $\Omega$  Schutzwiderstände. Diese bilden mit den Kondensatoren einen Tiefpaß (100k $\Omega$ /0,1 $\mu$ F) mit einer Grenzfrequenz von 16Hz. Dadurch werden hochfrequente Signale unterdrückt und die Spannung für Umschaltströme gepuffert. Sind die Meßsignale für eine Stromschleife aufbereitet, so können die Jumper J20 bis J27 100 $\Omega$  Widerstände parallel zu den Eingängen legen. Ein Eingangsstrom von 20mA erzeugt dann eine Meßspannung von 2V. Die Genauigkeit der Strommessung wird wesentlich von der Genauigkeit der 100 $\Omega$  Widerstände bestimmt. In der Standardausführung der Baugruppe sind hier 1% Fehler zu erwarten (1% Metallschichtwiderstände).

Der verwendete Multiplexer DG508 ist ein Industriestandard und läßt eine sehr fehlerarme Umschaltung von Signalen bis etwa +/-2,5V bei 5V Betriebsspannung zu. Bei höheren Eingangsspannungen nimmt der Durchlaßwiderstand und der Leckstrom der Analogschalter zu. Der Effekt ist auch bei 4,096V Eingangsspannung noch erträglich wenn das Eingangssignal niederohmig anliegt. Der Multiplexer ist nicht gegen Überspannungen (größer als die Betriebsspannung) am Eingang geschützt. Auf den Einsatz von Klemmdioden wurde wegen des Leckstromes verzichtet. Besteht die Gefahr, daß Eingangsspannungen größer als +/-5V auftreten, so sollte ein Multiplexer mit Eingangsschutz (fault-protected) verwendet werden. Die Bausteine MAX358 und HI508 beinhalten einen Schutz bis +/-35V, modernere Entwicklungen ertragen +/-100V Eingangsspannung.

Am Ausgang des Meßstellenumschalters folgt eine Anordnung aus RC-Tiefpaß und dem Jumperfeld ST2. Damit ergeben sich verschiedene Möglichkeiten für die Eingangsbeschaltung des AD-Umsetzers.

Betrieb mit Multiplexer:

Verwendet man den Multiplexer, so wird immer asymmetrisch (auf Masse bezogen) gemessen. J3 verbindet MUX und High-Eingang, J4 stellt den Massebezug des Low-Eingangs her. Der Jumper J6 ist entfernt, da schon Kondensatoren am Eingang des Multiplexers vorhanden sind. Anderenfalls müßte man nach jeder Umschaltung der Eingänge eine Meßpause von 0,9s einlegen, bis sich der Kondensator am Eingang des ADU umgeladen hat. Der Jumper J5 bestimmt den Eingangswiderstand der Schaltung zu etwa 10M $\Omega$ . Dies hat zur Folge, daß ein Spannungsteiler aus den 100k $\Omega$  Vorwiderständen und dem 10M $\Omega$  Abschlußwiderstand gegen Masse entsteht.

Die Eingangsspannung wird um den Faktor 0,9901 abgeschwächt. Die Toleranz der Vorwiderstände untereinander ist dabei bedeutungslos. Die Referenzspannung ist ebenfalls um etwa 1% niedriger eingestellt, als ohne den Abschlußwiderstand, um den Teilerfaktor wieder auszugleichen. Nach jeder Umschaltung auf einen anderen Kanal, muß eine Pause von 45ms dem Kondensator am jeweiligen Eingang ermöglichen, sich entsprechend dem Spannungsteilerverhältnis umzuladen. Dies ist jedoch die bevorzugte Beschaltung und so im Schaltplan eingezeichnet. Die Wartezeit von 45ms kann durch die Ausführungsgeschwindigkeit des MCS Basic-52 praktisch entfallen.

Alternativ kann der 10M $\Omega$  Widerstand abgetrennt werden. Der Eingangswiderstand liegt dann im Giga-Ohm-Bereich. Die Referenzspannung muß natürlich zuvor neu abgeglichen werden. Dann ist es auch möglich den 1M $\Omega$  Widerstand durch eine Steckbrücke zwischen Pin 8 und Pin 10 zu brücken, um die Ausgangsspannung des Multiplexers direkt an den Eingang des AD-Umsetzers zu verbinden. Dies bietet jedoch keinerlei Vorteile und ist deshalb auch nicht im Schaltbild vermerkt.

Betrieb ohne Multiplexer:

Werden die Jumper J3, J4 entfernt, so steht der "symmetrische" Eingang des ICL 7109 zu Verfügung. Die Symmetrie ist mit Vorsicht zu genießen; Die Gleichtaktspannung sollte immer 1,5V unter der Betriebsspannung liegen, hier also +/-3,5V. Beide Anschlüsse sollten innerhalb des Versorgungsspannungsbereichs liegen. Für eine asymmetrische Messung kann man den Low-Eingang mit dem Jumper J4 an Masse legen. Der Eingangswiderstand liegt im Giga-Ohm-Bereich. Sinnvollerweise wird J6 gesteckt und die Spannung über den 1M $\Omega$  Widerstand zugeführt. Die Spannung durchläuft dann einen Tiefpaß mit einer Zeitkonstante von  $\tau = 0,1s$ . Nach einem Spannungssprung sollte erst 0,9s später gemessen werden. Die Eingänge sind an den Stiften 4 (High) und 6 (Low) zu finden. Bei Bedarf kann mit J5 der 10M $\Omega$  Widerstand parallel zu den Klemmen gelegt werden, um den Eingangswiderstand zu definieren. Der Anschluß erfolgt an ST2 mit einem Pfostenstecker und Flachbandkabel. Anstelle der Jumper kann man die Verbindungen im Stecker löten oder die Adern des Kabels entsprechend verbinden. Ein Neuabgleich der Baugruppe (maximale Eingangsspannung) ist für diese Betriebsart in jedem Fall erforderlich.

Analog-Digital-Umsetzer:

Die Differenzeingangsspannung wird zunächst gepuffert und in einen proportionalen Strom ( $(U_{High} - U_{Low}) / R_{int}$ ) umgewandelt. Damit lädt das IC den Integrationskondensator  $C_{int}$  eine konstante Zeit von 40ms lang auf (2048 Takte). Die Kondensatorspannung ist danach proportional zum Mittelwert der Eingangsspannung. Durch die gewählten 40ms für die Integration heben sich Brummspannungen mit 25Hz, 50Hz, 75Hz, 100Hz usw. auf. Wechselspannungen werden um so stärker unterdrückt, je höher ihre Frequenz ist, und Frequenzen mit einem ganzzahligen Vielfachen von 1/40ms werden vollständig unterdrückt.

In der zweiten Phase wird mit Analogschaltern der Eingang des Integrators so umgeschaltet, daß der Kondensator durch die Referenzspannung bis auf Null Volt entladen wird. Bei einer negativen Eingangsspannung wird die Referenz umgepolt. Die Zeit der Entladung wird intern mit einem einfachen Zähler und einem Takt aus Quarzoszillator und Vorteiler ( $f_{takt} = 3MHz / 58$ ) gemessen. Der Zählerstand ist direkt proportional zur Eingangsspannung (maximal 4095 Takte).

Die Taktfrequenz bestimmt die Unterdrückung der Brummspannung, hat aber weiter keinen Einfluß auf die Meßgenauigkeit. Eine optimale Brummunterdrückung erreicht man nur mit einem 2,9696MHz Quarz. Der Fehler durch den verwendeten preiswerten 3MHz Quarz ist aber vernachlässigbar gering. Ebenso ist die Integrationszeitkonstante  $R_{int} * C_{int}$  in Grenzen bedeutungslos. Die Eingangsspannung wird nur durch das Verhältnis der Zählerstände bestimmt. (Für andere Meßbereiche ist jedoch eine Anpassung entsprechend der Tabelle im Schaltplan angebracht, um eine Sättigung oder Übersteuerung zu vermeiden.)

Der Umsetzer ICL 7109 kompensiert vor jeder Messung seinen Nullpunktfehler in einer sogenannten Auto-Zero-Phase. Dabei wird intern der Eingang kurzgeschlossen und der Kondensator  $C_{AZ}$  auf die Summe aller Offsetspannungen im System aufgeladen. Bei der nächsten Messung werden die Offsetspannungen mit der Kondensatorladung kompensiert. Ein Nullpunktgleich der Schaltung kann entfallen. Die Baugruppe ist also mit der Einstellung der maximalen Eingangsspannung komplett justiert.

#### Referenzspannung:

Die maximale Eingangsspannung beträgt  $U_{FS} = U_{ref} * 4095/2048$  entsprechend den Zählerkapazitäten. Üblicherweise ist hier  $U_{FS} = 2 * U_{ref}$  angegeben, dieser Wert (4096 Digit) kann jedoch niemals angezeigt werden. Der AD-Umsetzer ICL 7109 enthält bereits eine Referenzspannungsquelle, deren Temperaturstabilität jedoch nicht überzeugt (80ppm/°C). Die Referenzspannung wird hier mit einer integrierten Bandgapreferenz MX-, LT- oder AD 580 erzeugt. Der Ausgang liefert konstante 2,5V mit einer Toleranz von 40ppm/°C. Ausführungen mit 10, 25, 40 und 80ppm/°C Temperaturgang sind ebenso verfügbar. Die Spannung wird mit einem Zehngang-Trimmpoti geteilt, so daß sich für die Gesamtschaltung ein Abgleichbereich auf beliebige  $U_{FS} \pm 5V$  ergibt (naja). Günstig sind Eingangsspannungen von 4,096V ( $U_{ref} = 2,048V$ ,  $U_{LSB} = 1mV$ ), 2,048V ( $U_{ref} = 1,024V$ ,  $U_{LSB} = 500\mu V$ ) oder 0,4096V ( $U_{ref} = 0,2048V$ ,  $U_{LSB} = 100\mu V$ ), da keine großartige Umrechnung des Binärcodes mehr nötig ist.

#### Abgleich:

Der Abgleich kann mit dem angegebenen Basic-Programm erfolgen. Man läßt den Umsetzer kontinuierlich einen Kanal messen und schließt dort eine Spannung mit der Größe  $U_{FS} - 1,5 * U_{LSB}$  (z.B.: 4,096V - 1,5mV = 4,0945V) an. Die Referenzspannung wird nun so eingestellt, daß der Binärcode zwischen 4094 und 4095 schwankt. (Das ist schwer hinzukriegen, entscheiden Sie sich!)

Um den Einfluß aller möglichen Fehlerquellen gering zu halten, ist es ratsam, stets den größtmöglichen Eingangsspannungsbereich zu wählen. Dies gilt auch dann, wenn die Leckströme des Multiplexers dadurch zunehmen, da alle möglichen Offsetspannungen des Systems als konstante Größen auftreten. Besonders, wenn durch die geschickte Wahl des Eingangsbereiches ein Spannungsteiler entfallen kann, ist der Gewinn an Genauigkeit erheblich.

#### Layout:

Damit der Analog-Digital-Umsetzer genaue Ergebnisse liefert ist besondere Sorgfalt bei der Auslegung des Platinenlayouts notwendig. Die größten Fehler entstehen, wenn die Versorgungsströme des Analog- und des Digitalteils über die gleichen Leiterbahnen und Anschlüsse fließen und diese zusätzlich in der Bezugsleitung des Analogeingangs liegen. Der Mittelwert der Betriebsströme erzeugt einen Spannungsabfall, der als Offsetspannung direkt in das Meßergebnis eingeht. Diese Spannung kann auch mit dem Auto-Zero-Verfahren nicht kompensiert werden. Darüber hinaus sind die Ströme zeitlich nicht konstant, wodurch der Meßwert oszilliert und eine rechnerische Korrektur des Ergebnisses ausscheidet. Abhilfe schafft eine Anordnung mit getrennten Betriebsspannungsquellen für den Analog- und den Digitalteil, bei der es nur eine Ausgleichsleitung zwischen den analogen- und digitalen Bezugsanschlüssen gibt. Zusätzlich darf die Verbindung zwischen Low-Eingang und Analogmasse keinen Betriebsstrom führen. Auch die Referenzspannungsmasse sollte direkt an den analogen Bezugspunkt geführt sein. Ferner ist eine kapazitive Verkopplung zwischen den digitalen Steuerleitungen und den Bauelementen des Analogteils zu vermeiden.

Die Forderung nach getrennten Betriebsspannungen ließ sich hier nicht erfüllen. Der ECB-Bus (wie auch andere Bussysteme) sieht nur eine Masseleitung vor, und die Schaltung sollte mit einer einfachen Versorgung von 5V arbeiten. Die Verwendung von DC/DC-Wandlern schied aus Kostengründen und wegen der zu erwartenden Störfelder aus. Die einzige mögliche Abhilfe war eine getrennte Masseführung auf der Platine. Der analoge Bezugspunkt liegt direkt an dem entsprechenden Pin des AD-Umsetzers. Hier sind der Low-Eingang, die Tiefpaßkondensatoren und die Referenzmasse angeschlossen. Von dort führt eine Leitung an den gemeinsamen Masseanschluß der Busverbindung. Diese Verdrahtung liefert die optimalen Ergebnisse.

Alle Betriebsspannungsanschlüsse sind, direkt am ADU, mit einem Elko und einem keramischen Kondensator abgeblockt. Die negative Hilfsspannung wird mit einer integrierten Ladungspumpe erzeugt. Da die negativen Betriebsströme des Analog-Digital-Umsetzers und des Multiplexers zusammen weniger als 4 mA betragen, reicht diese einfache Schaltung zur Versorgung aus. Die negative Spannung durchläuft zusätzlich ein RC-Glied um Störungen zu dämpfen.

#### Steuerung der ADU:

Das Ergebnis der AD-Umsetzung steht an den Datenleitungen parallel zur Verfügung. Das IC wird hier im Direktmodus (Mode = 0) betrieben, so daß die Steuerung des Datentransfers mit den Leitungen /LBEN, /HBEN und /CE erfolgt. Zum Auslesen wird mit der /CE-Leitung der Baustein freigegeben. Mit den Adreßleitungen A0 und A1 werden nacheinander das niederwertige und das höherwertige Byte auf den Datenbus geschaltet. A0 = 0 (/LBEN) aktiviert das Low-Byte, also D0 bis D7. Mit A1 = 0 (/HBEN) werden die Daten D8 bis D11 auf die Busleitungen D0 bis D3 und das Overflow- und Polaritätsbit auf D6 und D7 gelegt. In diesem Fall muß der Rechner die Bits D4 und D5 ausmaskieren, sie sind nicht definiert. Die Adressen sind weiter unten nochmal aufgeführt.

#### Kontinuierliche Messung:

Die Steuerung der Umsetzung erfolgt mit der R/H-Leitung (Run/Hold). Dies ist ein Eingang des ADUs und wird vom Port C7 des Peripheriebausteins 8255 gesteuert. Führt die Leitung Eins-Pegel, so wird kontinuierlich eine Umsetzung nach der anderen gestartet. Die Umsetzdauer beträgt dabei 160ms, und das Ergebnis ist zwischenzeitlich abzufragen. Eine Rückmeldung erfolgt mit dem Status-Signal STAT auf den Port C0 des 8255. Ist STAT = 0, so ist die Umsetzung abgeschlossen. Der ADU befindet sich dann in der Auto-Zero-Phase, und die Daten können aus dem internen Latch gelesen werden. Dafür stehen etwa 40 bis 118ms zur Verfügung, je nach Größe der Eingangsspannung. Während der Messung liegt STAT auf logisch Eins und die Daten sind ungültig. Bei dieser Methode ist sicher zu stellen, daß STAT beim Auslesen der Daten nicht zwischendurch den Pegel wechselt. Am besten ist testen; vorher und nachher.

#### Einzelwertmessung:

In dem angegebenen Beispielprogramm wird etwas anders verfahren: Ein kurzer Eins-Impuls auf der R/H-Leitung startet eine Umsetzung. STAT springt nun auf High-Pegel (mit 0,135ms Verzögerung) und fällt wieder auf Low zurück wenn die Umsetzung abgeschlossen ist (40 bis 120ms später). Die Daten sind jetzt gültig, und der Vorgang kann bei Bedarf neu gestartet werden. Der nächste Startimpuls darf erst gesendet werden, nachdem die Auto-Zero-Phase abgeschlossen ist, also mindestens 40ms nach der letzten Umsetzung. Der Eins-Impuls soll länger als 200ns (höhö) und kürzer als 160ms sein. Ein zweiter Impuls während der Umsetzung oder der Auto-Zero-Phase bewirkt nichts. Die gesamte Umsetzdauer ist abhängig von der Größe der Eingangsspannung und beträgt 80ms für Null Volt und 160ms bei maximaler Spannung.

#### Digitale Ein-/Ausgänge:

Der verwendete Peripheriebaustein 8255 übernimmt außer der Steuerung der AD-Umsetzung noch weitere Aufgaben. Der Baustein wird hier im Modus 0 betrieben, d.h. die Ports arbeiten als einfache Ein- und Ausgabegruppen mit fester Datenrichtung. Dabei läßt sich der Port C in eine obere und untere Hälfte (Nibble) mit unterschiedlicher Richtung aufteilen. Das High-Nibble des Port C (C4 bis C7) ist als Ausgang zu verwenden. Dabei steuert C7 die Run/Hold-Leitung, wie oben beschrieben. Die Portanschlüsse C4, C5 und C6 sind die Auswahlleitungen des Multiplexers. Entsprechend der eingestellten Bitkombination (000 bis 111) ist der Analogeingang (1 bis 8) aktiviert. Das Low-Nibble des Port C wird als Eingang programmiert. C0 ist mit der Statusleitung des ADU verbunden. Die Portleitungen C1, C2 und C3 sind als Eingänge frei verwendbar und liegen auf der Pfostenleiste ST3.

Der Port B des 8255 ist mit acht Jumpern J10 bis J17 verbunden, um ein Konfigurationsbyte einzustellen. Diese Jumperleiste liegt auf der Platine parallel zu den Jumpern J20 bis J27 für die Stromeingänge. Es sollten hier die Jumper gesteckt werden, die gerade keine Stromsenken aktivieren. Dann kann per Programm ausgelesen werden, welche Eingänge Strom und Spannung messen:

- Verbindet der Jumper den Widerstand mit dem Eingang, so wird Strom gemessen, und ein Lesezugriff auf Port B liefert für dieses Bit eine Eins.
- Wird eine Spannung gemessen, so steckt der Jumper am Port B, und ein Lesezugriff liefert für diese Bit eine Null.

Mit dem verbleibenden Port A steuert der I/O-Baustein acht, über Optokoppler getrennte, Darlingtonttransistoren an. Damit steht ein potentialfreier Ausgangsport zur Verfügung. Externe Spannungen bis 30V und jeweils 500mA pro Ausgang (Summe 6 2A) können geschaltet werden. Die Betriebsspannung ist von außen an die Anschlüsse 11 (+) und 9 oder 10 (-) der D-Sub-Buchse anzulegen.

**Achtung: Die Kriech- und Isolationsstrecken sind sehr gering und nicht für eine Netzspannungstrennung geeignet!**

Setzt man die Jumper J7 und J8 an der Stiftleiste ST6, so wird die Betriebsspannung des Bussystems (+5V) geschaltet. Die Ausgänge sind dann natürlich nicht mehr potentialfrei. Der maximale Laststrom ist bei 5V-Betriebsspannung erheblich kleiner als 500mA. Wenn dies stört sind die Widerstände zwischen Optokoppler und des Transistorbasen zu verkleinern. Möchte man die Ausgangsbeschaltung ändern, so stehen alle Anschlüsse an den Pfostenleisten ST3 und ST6 bereit.

#### Busanschaltung:

Die Datenleitungen werden mit einem bidirektionalen Bustreiber gepuffert. Die Datenrichtung wird dabei mit der /ReaD-Leitung erkannt. Als Adreßdeko-der ist ein GAL-Baustein 16V8 eingesetzt, so daß auf Wunsch eine neue Adreßbelegung programmiert werden kann. Im IO-Bereich des Bussystems sind die Leitungen A0 bis A7 ausdekodiert. Der Memory-Bereich wird nicht benutzt. Die Baugruppe belegt acht aufeinanderfolgende Adressen. Die Basisadresse ist mit den drei Jumpern J0 bis J2 im Bereich 90h..C8h einstellbar. Der GAL-Baustein erzeugt zusätzlich das RESET-Signal für den Peripheriebaustein.

.. Adreßbelegung ADI-12 -----  
 |  
 -----

Die Basisadresse wird mit den Jumpern J0 bis J2 eingestellt:

J2	J1	J0	Basisadresse	
0	0	0	90h	Jx = 0 entspricht
0	0	1	98h	Jumper gesetzt.
0	1	0	0A0h	
0	1	1	0A8h	Jx = 1 entspricht
1	0	0	0B0h	Jumper entfernt.
1	0	1	0B8h	
1	1	0	0C0h	Für den Rechner BasiControl ist der
1	1	1	0C8h	Basisadresse 0E000h hinzuzuaddieren.

Die Adressen einer Baugruppe sind wie folgt belegt:

Basisadr.+	Funktion	Datenrichtung	Bemerkung
0	Port A (8255)	schreiben	Digitale Ausgänge
1	Port B	lesen	Jumperleiste
2	Port C	lesen/schreiben	Steuerung des ADU
3	Steuerwortreg.	lesen/schreiben	siehe unten
4	verboten!!!	-	A0 = 0 und A1 = 0
5	High-Byte ADU	lesen	siehe unten
6	Low-Byte ADU	lesen	
7	nicht belegt	-	A0 = 1 und A1 = 1

Bei der vorgesehenen Beschaltung ergibt sich folgendes Steuerwort:

Port A = Ausgang                    Port C Low-Nibble = Eingang  
 Port B = Eingang                    Port C High-Nibble = Ausgang

=> resultierendes Steuerwort = 83h

Das Steuerwort ist am Anfang des Programms in das Steuerwortregister (93h) zu schreiben. Es kann auch gelesen werden. Nach dem Einschreiben sind die Ausgangsleitungen des Port A logisch Null, der Ausgang also aktiv. Beim Schreiben und Lesen von Port C ist jeweils nur das obere oder untere Nibble gültig. Die Ausgangsleitungen von Port C können auch einzeln über das Steuerwortregister gesetzt werden. Das Byte ergibt sich zu:

0 x x x D3 D2 D1 D0      x = egal  
 .----- .----- = Wert (0/1)  
 .----- = Bit-Nummer (hier C4 bis C7)

Die Adresse 94h darf nicht benutzt werden. Die Adreßleitungen A0 und A1 sind beide Null, so daß der AD-Umsetzer Low und High-Byte gleichzeitig ausgibt. Bei der Adresse 97h geschieht nichts, da der Umsetzer weder Low- noch High-Byte ausgibt.

Die Daten sind wie folgt den Leitungen zugeordnet:

Datenleitung	D0	D1	D2	D3	D4	D5	D6	D7	
Low-Byte ADU	D0	D1	D2	D3	D4	D5	D6	D7	POL = 1 >> U <sub>e</sub> positiv
High-Byte ADU	D8	D9	D10	D11	"1"	"1"	OR	POL	OR = 1 >> U <sub>e</sub> > U <sub>FS</sub>



```
..- Demoprogramm -----  
|                                                                     |  
-----
```

Das folgende Programm soll anhand eines Beispiels die Programmierung der Baugruppe ADI-12 erläutern.

Im Programm wird zunächst der Peripheriebaustein initialisiert und danach im Hauptmenue die gewünschte Funktion erfragt. Der Ablauf verzweigt in die Unterprogramme: Digitale Ausgabe über Port A, Einlesen der Jumper am Port B, Analog-Digital-Umsetzung eines Kanals und AD-Umsetzung aller Kanäle nacheinander. Vor jeder Messung wird zunächst der Multiplexer gesetzt und der ADU angehalten. Dies wird mit der Statusleitung überprüft. Ein kurzer Startimpuls leitet danach eine neue Messung ein. In einer DO-UNTIL-Schleife wird wiederum die Statusleitung geprüft. Ist die Umsetzung abgeschlossen, wird nacheinander das Low- und das High-Byte ausgelesen. Schließlich wertet das Programm noch die Vorzeichen- und Overflow-Bits aus und formatiert die Ergebnisse passend. Das Programm ist nicht zeitoptimiert. Das Statusbit wird solange eingelesen und geprüft, bis die Umsetzung abgeschlossen ist. In dieser Zeit könnten in einer wirklichen Anwendung auch andere Aufgaben erledigt werden.

Es sind nacheinander der Quellcode für den "Compreter52" und das Programm für den Intel-Basicinterpreter im 8052AH-BASIC abgedruckt.

Im Compreter-Quellcode bedeuten:

```
; ' und /*      Ein Kommentar folgt. Im Gegensatz zum normalen Basic-Kommentar  
(REM...) wird hier der Text beim Download unterdrückt.
```

```
[INSTEAD_OF      Ein Makro mit nachfolgender Definition. Der Compreter setzt im  
Programm an Stelle des Makros jedesmal die Makrodefinition ein.
```

```
~GO_FOR_GOLD     Ein Label (Sprungziel). Der Name wird beim Download in eine  
Zeilennummer übersetzt.
```



```

~DIG_OUT    REM Menue digitale Ausgabe
PRINT: PRINT
PRINT " Geben Sie das auszugebende Byte an. Werte größer 0FFh brechen"
INPUT " den Vorgang ab: ",dw
IF dw > 0FFh THEN RETURN          /* Abbruchbedingung prüfen
dw = 0FFh .AND. NOT (dw)          /* Invertieren
[ADI_A_OUT = dw                   /* und ausgeben
PRINT
PRINT
GOTO ~DIG_OUT                      /* auch hier endlos
-----
~JUMP_IN    REM Einlesen der Jumper
PRINT: PRINT
PRINT " Jumper gesteckt = 1 (Spannungsmessung),"
PRINT " Jumper offen = 0 (Strommessung)"
PRINT " Weiter mit beliebiger Taste": PRINT
PRINT USING(#), " J7 J6 J5 J4 J3 J2 J1 J0" /* Tabelle anzeigen
DO                                          /* endlos aktuell
    jump = 0FFh .AND. NOT ([ADI_JUMP])    /* einlesen, negieren
    b0 = jump .AND. 01h                   /* 0/1 erzeugen
    b1 = jump .AND. 02h: b1 = b1/02h
    b2 = jump .AND. 04h: b2 = b2/04h
    b3 = jump .AND. 08h: b3 = b3/08h
    b4 = jump .AND. 10h: b4 = b4/10h
    b5 = jump .AND. 20h: b5 = b5/20h
    b6 = jump .AND. 40h: b6 = b6/40h
    b7 = jump .AND. 80h: b7 = b7/80h
    PRINT CR, " ", b7, b6, b5, b4, b3, b2, b1, b0,
    key = GET                             /* irgendwas einlesen
UNTIL key <> 0                             /* Abbruch wenn Taste
PRINT USING(0)                             /* normales Format
RETURN                                       /* Hauptmenue
-----
~ADU        REM AD-Umsetzung
PRINT: PRINT
PRINT " Analog-Digital-Umsetzung eines Kanals"
~AD2        PRINT " Geben Sie bitte die Kanalnummer (0..7) an. Werte größer"
PRINT " als 7 brechen die Umsetzung ab: "
INPUT " ", kanal                          /* Wert holen
IF kanal > 7 THEN RETURN                  /* Abbruchbedingung
PRINT
PRINT " Messung Kanal Nr.", kanal, ". Abbruch mit beliebiger Taste."
PRINT USING(##.####)                    /* optimal für 2,0475V
[ADI_MUX kanal                          /* Multiplexer setzen und HOLD
DO                                        /* warten ?
    busy = [ADI_STAT                     /* 1 = beschäftigt
UNTIL busy = 0
DO                                        /* evtl. überflüssig
    PRINT CR, " ",                       /* Formatierung
    [ADI_RUN: [ADI_HOLD                   /* Startpuls
    DO                                    /* wie lange dauert es?
        busy = [ADI_STAT
        PRINT ">",                       /* mal schauen
    UNTIL busy = 0                       /* Umsetzung fertig
    low = [ADI_LOW                        /* Low-Byte holen
    high = [ADI_HIGH                      /* High-Byte holen
    IF (high .AND. 40h) > 0 THEN PRINT "too much !!!!",
    IF (high .AND. 80h) = 0 THEN PRINT " -", ELSE PRINT " +",
    al = 100h * (0Fh .AND. high) + low
    PRINT al/[ADI_FAKT,"V                ", /* alles einzeilig
    key = GET
UNTIL key <> 0                             /* neuer Kanal wenn Taste
PRINT: PRINT USING(0)                     /* Format zurücksetzen
GOTO ~AD2                                  /* neuen Kanal wählen
-----
~ADZ        REM AD-Umsetzung aller Kanäle nacheinander
PRINT: PRINT

```



Das Programm wurde übersetzt und ergab den folgenden Code für den Interpreter (MCS BASIC-52). Die Zeilennummern entsprechen den Zeilen im Compreterprogramm:

```
1    REM ADI_12.DOC
35   XBY(0E098h+3) = 83h
36   XBY(0E098h) = 0FFh
38   REM Hauptmenue
39   PRINT: PRINT: PRINT
40   PRINT " ADI-12 Testprogramm"
41   PRINT
42   PRINT " 8-Bit digitale Ausgabe      = 1"
43   PRINT " Jumper einlesen             = 2"
44   PRINT " Analogwert einlesen         = 3"
45   PRINT " Kanalele zyklisch einlesen  = 4"
46   PRINT " Ende mit Control-C oder     = 5"
47   PRINT
48   INPUT wahl
49   PRINT
50   IF wahl = 1 GOSUB 57
51   IF wahl = 2 GOSUB 68
52   IF wahl = 3 GOSUB 90
53   IF wahl = 4 GOSUB 122
54   IF wahl = 5 GOTO 156
55   GOTO 38
57   REM Menue digitale Ausgabe
58   PRINT: PRINT
59   PRINT " Geben Sie das auszugebende Byte an. Werte groesser 0FFh brechen"
60   INPUT " den Vorgang ab: ",dw
61   IF dw > 0FFh THEN RETURN
62   dw = 0FFh .AND. NOT (dw)
63   XBY(0E098h) = dw
64   PRINT
65   PRINT
66   GOTO 57
68   REM Einlesen der Jumper
69   PRINT: PRINT
70   PRINT " Jumper gesteckt = 1 (Spannungsmessung),"
71   PRINT " Jumper offen   = 0 (Strommessung)"
72   PRINT " Weiter mit beliebiger Taste": PRINT
73   PRINT USING(#), " J7 J6 J5 J4 J3 J2 J1 J0"
74   DO
75   jump = 0FFh .AND. NOT (XBY(0E098h+1))
76   b0 = jump .AND. 01h
77   b1 = jump .AND. 02h: b1 = b1/02h
78   b2 = jump .AND. 04h: b2 = b2/04h
79   b3 = jump .AND. 08h: b3 = b3/08h
80   b4 = jump .AND. 10h: b4 = b4/10h
81   b5 = jump .AND. 20h: b5 = b5/20h
82   b6 = jump .AND. 40h: b6 = b6/40h
83   b7 = jump .AND. 80h: b7 = b7/80h
84   PRINT CR, " ", b7, b6, b5, b4, b3, b2, b1, b0,
85   key = GET
86   UNTIL key <> 0
87   PRINT USING(0)
88   RETURN
90   REM AD-Umsetzung
91   PRINT: PRINT
92   PRINT " Analog-Digital-Umsetzung eines Kanals"
93   PRINT " Geben Sie bitte die Kanalnummer (0..7) an. Werte groesser"
94   PRINT " als 7 brechen die Umsetzung ab: "
95   INPUT " ", kanal
96   IF kanal > 7 THEN RETURN
97   PRINT
98   PRINT " Messung Kanal Nr.", kanal, ". Abbruch mit beliebiger Taste."
99   PRINT USING(#####)
100  XBY(0E098h+2) = 16 * kanal
101  DO
```

```

102 busy = 01h .AND. XBY(0E098h+2)
103 UNTIL busy = 0
104 DO
105 PRINT CR," ",
106 XBY(0E098h+3) = 0Fh: XBY(0E098h+3) = 0Eh
107 DO
108 busy = 01h .AND. XBY(0E098h+2)
109 PRINT ">",
110 UNTIL busy = 0
111 low = XBY(0E098h+6)
112 high = XBY(0E098h+5)
113 IF (high .AND. 40h) > 0 THEN PRINT "too much !!!!",
114 IF (high .AND. 80h) = 0 THEN PRINT " -", ELSE PRINT " +",
115 al = 100h * (0Fh .AND. high) + low
116 PRINT al/1000,"V          ",
117 key = GET
118 UNTIL key <> 0
119 PRINT: PRINT USING(0)
120 GOTO 93
122 REM AD-Umsetzung aller Kanäle nacheinander
123 PRINT: PRINT
124 PRINT " Analog-Digital-Umsetzung zyklisch"
125 PRINT " Abbruch mit beliebiger Taste."
126 PRINT
127 DO
128 FOR kanal = 0 TO 7
129 XBY(0E098h+2) = 16 * kanal
130 DO
131 busy = 01h .AND. XBY(0E098h+2)
132 UNTIL busy = 0
134 PRINT kanal, ": ",
135 XBY(0E098h+3) = 0Fh: XBY(0E098h+3) = 0Eh
136 DO
137 busy = 01h .AND. XBY(0E098h+2)
138 UNTIL busy = 0
139 low = XBY(0E098h+6)
140 high = XBY(0E098h+5)
141 IF (high .AND. 40h) > 0 THEN PRINT "too much !!!!",
142 IF (high .AND. 80h) = 0 THEN PRINT " -", ELSE PRINT " +",
143 al = 100h * (0Fh .AND. high) + low
144 PRINT USING(#####),
145 PRINT al/1000,"V ",
147 PRINT USING(0),
148 IF kanal = 3 THEN PRINT
149 NEXT kanal
150 PRINT: PRINT
151 key = GET
152 UNTIL key <> 0
153 PRINT: PRINT
154 RETURN
156 END

```

.. Fehlergrenzen und Meßgenauigkeit -----  
 |  
 |  
 -----

Die Baugruppe wurde verglichen mit zwei 4 1/2 stelligen Handmultimetern M-4650 CR (Gerät A) und M-4630 (Gerät B) der Firma Metex. Die Geräte haben jeweils einen Meßbereich von 0,2V mit 10uV Auflösung, 2V mit 100uV Auflösung und 20V mit 1mV Auflösung. Der maximale Fehler ist für beide Geräte angegeben mit +/-0,05% vom Anzeigewert +3 Digit. Der Eingangswiderstand beträgt 10M $\Omega$ . Als Spannungsquelle diente ein potentialfreies Netzgerät mit geringer Rausch- ( $U_{R_{SS}} = 1mV$ ) und Brummspannung ( $U_{BR_{SS}} = 3mV$ ). Die Ausgangsspannung von 4,1V wurde mit einem 10k $\Omega$  10-Gang-Poti auf die gewünschten Werte geteilt. Ein 100 $\Omega$  Vorwiderstand, ebenfalls ein 10-Gang-Poti, diente zur Feineinstellung. Beide Meßgeräte waren mit kurzen Drähten parallel zu den Eingängen 1 und 4 der Baugruppe verschaltet. Der Eingang 7 blieb unbeschaltet, Eingänge 2, 3, 5, 6 und 8 waren mit dem Masseanschluß der Eingangsbuchse verbunden.

Die Brücken waren entsprechend dem Schaltbild gesteckt. Der Meßbereichsendwert wurde zu 4,096V gewählt. Der Abgleich der Referenzspannung wurde aber bei 1,0000V (+/-400uV) Anzeige am Gerät A vorgenommen (Dies hat den Vorteil, daß die Auflösung des Multimeters um eine Stelle größer ist als bei 4,096V).

Messung nur am Eingang 1 der ADI-12:

ADI-12/V	dU/V	Gerät A/V	Gerät B/V	Fehlergrenzen A, B
+4,095	+4,0mV	+4,091	+4,094	+/-2mV +3mV
+4,094	+4,0mV	+4,090		+/-2mV +3mV
+3,503	+3,0mV	+3,500	+3,504	+/-2mV +3mV
+2,503	+3,0mV	+2,500	+2,503	+/-1mV +3mV
+1,992	+1,2mV	+1,9908	+1,9931	+/-1,0mV +0,3mV
+1,991	+0,6mV	+1,9904	+1,9927	+/-1,0mV +0,3mV
+1,500	0,0mV	+1,5000	+1,5018	+/-0,8mV +0,3mV
+1,000		+1,0004		+/-0,5mV +0,3mV
+1,000	0,0mV	+1,0000	+1,0016	+/-0,5mV +0,3mV
+1,000		+0,9996		+/-0,5mV +0,3mV
+0,500	0,0mV	+0,5000	+0,5009	+/-0,3mV +0,3mV
+0,195	+0,4mV	+0,194.60		+/-0,10mV +0,03mV
+0.194	-0,6mV	+0,194.40		+/-0,10mV +0,03mV
0,000	0,0mV	0,000.00	0,000.10	+/-0,00mV +0,03mV
-0,195	0,0mV	-0,195.00		+/-0,10mV +0,03mV
-0,500	-0,4mV	-0,4996		+/-0,3mV +0,3mV
-0,500	+0,6mV	-0,5006		+/-0,3mV +0,3mV
-1,000	0,0mV	-1,0000		+/-0,5mV +0,3mV
-1,500	0,0mV	-1,5000		+/-0,8mV +0,3mV
-1,999	-0,5mV	-1,9985	-1,9997	+/-1,0mV +0,3mV
-2,000	-1,2mV	-1,9988	-o.r.	+/-1,0mV +0,3mV
-2,503	-3,0mV	-2,500	-2,498	+/-1mV +3mV
-3,504	-4,0mV	-3,500	-3,497	+/-2mV +3mV
-4,094	-3,0mV	-4,091		+/-2mV +3mV
-4,095	-3,0mV	-4,092	-4,088	+/-2mV +3mV

Bei näherer Betrachtung der Ergebnisse fällt auf, daß die Genauigkeit der verschiedenen Geräte durchaus vergleichbar ist! D.h. die Abweichungen liegen im Toleranzbereich der Handmultimeter. (Die Angabe des Fehlers mit +/-0,05% +3Digit ist vermutlich bei negativen Eingangsspannungen mit negativem Vorzeichen zu interpretieren.)

Kontrollmessungen mit einer Steckbrücke zwischen Pin 10 und Pin 8 (ST2), ohne 10M $\Omega$  Widerstand, ergaben Meßwerte, die genau um den errechneten Faktor von 1,01 über den oben angegebenen lagen.

Mit dem abgedruckten Basicprogramm wurde eine weitere Meßreihe aufgenommen, bei der die Eingänge zyklisch abgetastet wurden. So sollte das Einschwingverhalten untersucht und beurteilt werden. Zunächst die Schaltung mit 10M $\Omega$  Abschlußwiderstand des Multiplexers:

Kanal	:	1	2	3	4	5	6	7	8
Ue/V	:	4,000	0	0	4,000	0	0	offen	0
Anz./V	max:	4,005	0	0	4,005	0	0	0,220	0
	min:	4,004	0	0	4,004	0	0	0,020	0

Einzelmessung nur Kanal 1: 4,000V                      Abweichung: 4..5 Digit

Messung mit Verbindung Pin 10 und Pin 8 (ST2), ohne 10M $\Omega$  Abschluß des Multiplexers:

Kanal	:	1	2	3	4	5	6	7	8
Ue/V	:	4,000	0	0	4,000	0	0	offen	0
Anz./V	max:	4,045	0	0	4,045	0	0	0,500	0
	min:	4,044	0	0	4,044	0	0	0,400	0

Einzelmessung nur Kanal 1: 4,040V                      Abweichung: 4..5 Digit

Erstaunlicherweise zeigt sich auch ohne Abschlußwiderstand des Multiplexers ein Einschwingverhalten der gleichen Größenordnung. Es wurden noch weitere Tests ohne die Vorwiderstände, mit und ohne 10M $\Omega$  Abschluß durchgeführt, die immer das gleiche Ergebnis lieferten. Ein Zusammenhang mit der Eingangsschaltung läßt sich nicht herstellen. Allein die Verwendung des Multiplexers, genauer die sprunghafte Änderung der Eingangsspannung und damit der aufeinanderfolgenden Meßwerte, ruft diesen Effekt hervor. Die Meßwerte sind betragsmäßig zu groß, was auch für negative Eingangsspannungen gilt. Evtl. kann dieser Effekt auf das Verhalten des Integrationskondensators zurückgeführt werden. Daraus folgt, das man nach dem Umschalten der Kanäle besser einige Male ins Leere mißt oder den Fehler per Software und Abgleich korrigiert.



.. Steckverbindungen ADI-12 -----  
 |  
 |  
 -----

ECB-Bus, 64polige VG-Leiste ST1:

Pin:	Funktion:	Pin:	Funktion:	
1c	+5V	1a	+5V	
2c	D0	2a	D5	es sind nur die
3c	D7	3a	D6	benutzten Anschlüsse
4c	D2	4a	D3	aufgeföhrt.
5c	A0	5a	D4	
6c	A3	6a	A2	
7c	A1	7a	A4	Die a-Reihe befindet
		8a	A5	sich an der Außenseite
9c	A7	9a	A6	der Platine.
14c	D1			
22c	/WR			
24c	/RD			
26c	PCL			
		27a	/IORQ	
32c	GND	32a	GND	

Pfostenverbinder am AD-Umsetzer-Eingang ST2:

Pin:	Funktion:	Jumper:
1,2	nicht belegt	
3	Ausgang MUX	--
4	Eingang Filter	-- J3
5	Analog-Masse	--
6	Low-Eingang ADU	-- J4
7	Abschlußwiderstand	--
8	Eingang Filter	-- J5
9	Filterkondensator	--
10	High-Eingang ADU	-- J6

Pfostenverbinder am Peripheriebaustein ST3:

Pin:	Funktion:	Pin:	Funktion:
1, 2	+5V	19, 20	Port C1
3, 4	Port A3	21, 22	" C2
5, 6	" A2	23, 24	" C3
7, 8	" A1	25, 26	GND
9, 10	" A0		
11, 12	" A4		
13, 14	" A5		
15, 16	" A6		
17, 18	" A7		

Analogeingang, 9poliger D-Sub-Stecker ST4:

Pin:	Funktion:
1	Analogeingang 1
2	Analogeingang 2
3	Analogeingang 3
4	Analogeingang 4
5	Analogeingang 5
6	Analogeingang 6
7	Analogeingang 7
8	Analogeingang 8
9	Analog-Masse

Digitaler Ausgang, 15polige D-Sub-Buchse ST5 und Pfostenverbinder ST6:

ST5-Pin:	ST6-Pin:	Funktion:	
1	17, 18	digitaler Ausgang A0	
2	15, 16	digitaler Ausgang A1	Invertierte Ausgänge
3	13, 14	digitaler Ausgang A2	von Port A des 8255-
4	11, 12	digitaler Ausgang A3	Peripheriebaustein.
5	9, 10	digitaler Ausgang A4	
6	7, 8	digitaler Ausgang A5	
7	5, 6	digitaler Ausgang A6	
8	3, 4	digitaler Ausgang A7	
9, 10	19	extern GND Eingang	--
-	20	intern GND	-- J8
11	1	extern +Ub Eingang (+5V..+30V)	--
-	2	intern +Ub (+5V)	-- J7
12	-	nicht belegt	
13	-	nicht belegt	
14	-	nicht belegt	
15	-	nicht belegt	

..- **Jumperfunktionen ADI-12** -----  

- J0 Basisadresse +08h wenn Jumper gezogen
- J1 Basisadresse +10h wenn Jumper gezogen
- J2 Basisadresse +20h wenn Jumper gezogen
  
- J3 Verbindung zwischen High-Eingang des AD-Umsetzers und Multiplexer
- J4 Verbindung zwischen Low-Eingang des AD-Umsetzers und Analog-Masse
- J5 10M $\Omega$  Abschlußwiderstand parallel zum Eingang
- J6 Filterkondensator für den Betrieb ohne Multiplexer
  
- J7 Digitaler Ausgang arbeitet mit interner Betriebsspannung (+5V)
- J8 Digitaler Ausgang arbeitet mit interner Masse
  
- J10 Konfigurationsbit B0 = 0            Port B des Peripherie-  
    baustein 8255.
- J11            "            "            B1 = 0
- J12            "            "            B2 = 0
- J13            "            "            B3 = 0
- J14            "            "            B4 = 0
- J15            "            "            B5 = 0
- J16            "            "            B6 = 0
- J17            "            "            B7 = 0
  
- J20 Stromsenke an Analogeingang 1
- J21 Stromsenke an Analogeingang 2
- J22 Stromsenke an Analogeingang 3
- J23 Stromsenke an Analogeingang 4
- J24 Stromsenke an Analogeingang 5
- J25 Stromsenke an Analogeingang 6
- J26 Stromsenke an Analogeingang 7
- J27 Stromsenke an Analogeingang 8

```

.- GAL-Dokumentation ADI-12 -----
|
|-----

```

```

Filename : ADI_12.LCI      PCB      : ADI_12
IC-Nr.   : -              Version  : 1.0          vom : 13.4.94

```

```

%ID      ADI_12
%TYP     GAL16V8

```

```
%PINS
```

```

'PIN 1   2   3   4   5   6   7   8   9   10 '
      /PCL /RD  /IORQ A7  A6  A5  A4  A2  A3  'GND'

'PIN 11  12  13  14  15  16  17  18  19  20 '
      A0  A1  /BE  RESET /CSP /CSA S2  S1  S0  'VCC'

```

```
%LOGIC
```

```

'Adresse = 90..97h + 08h*S0 + 10h*S1 + 20h*S2
'Sn = 1 wenn der Jumper offen ist

```

```
'Freigabe des Bustreibers:
```

```

BE = IORQ * A7 * A6 * !A5 * !A4 * A3 * S2 * S1 * S0
   + IORQ * A7 * A6 * !A5 * !A4 * !A3 * S2 * S1 * !S0
   + IORQ * A7 * !A6 * A5 * A4 * A3 * S2 * !S1 * S0
   + IORQ * A7 * !A6 * A5 * A4 * !A3 * S2 * !S1 * !S0
   + IORQ * A7 * !A6 * A5 * !A4 * A3 * !S2 * S1 * S0
   + IORQ * A7 * !A6 * A5 * !A4 * !A3 * !S2 * S1 * !S0
   + IORQ * A7 * !A6 * !A5 * A4 * A3 * !S2 * !S1 * S0
   + IORQ * A7 * !A6 * !A5 * A4 * !A3 * !S2 * !S1 * !S0;

```

```
'8255-CS:
```

```

CSP = IORQ * A7 * A6 * !A5 * !A4 * A3 * !A2 * S2 * S1 * S0
     + IORQ * A7 * A6 * !A5 * !A4 * !A3 * !A2 * S2 * S1 * !S0
     + IORQ * A7 * !A6 * A5 * A4 * A3 * !A2 * S2 * !S1 * S0
     + IORQ * A7 * !A6 * A5 * A4 * !A3 * !A2 * S2 * !S1 * !S0
     + IORQ * A7 * !A6 * A5 * !A4 * A3 * !A2 * !S2 * S1 * S0
     + IORQ * A7 * !A6 * A5 * !A4 * !A3 * !A2 * !S2 * S1 * !S0
     + IORQ * A7 * !A6 * !A5 * A4 * A3 * !A2 * !S2 * !S1 * S0
     + IORQ * A7 * !A6 * !A5 * A4 * !A3 * !A2 * !S2 * !S1 * !S0;

```

```
'7109-CS:
```

```

CSA = IORQ * RD * A7 * A6 * !A5 * !A4 * A3 * A2 * S2 * S1 * S0
     + IORQ * RD * A7 * A6 * !A5 * !A4 * !A3 * A2 * S2 * S1 * !S0
     + IORQ * RD * A7 * !A6 * A5 * A4 * A3 * A2 * S2 * !S1 * S0
     + IORQ * RD * A7 * !A6 * A5 * A4 * !A3 * A2 * S2 * !S1 * !S0
     + IORQ * RD * A7 * !A6 * A5 * !A4 * A3 * A2 * !S2 * S1 * S0
     + IORQ * RD * A7 * !A6 * A5 * !A4 * !A3 * A2 * !S2 * S1 * !S0
     + IORQ * RD * A7 * !A6 * !A5 * A4 * A3 * A2 * !S2 * !S1 * S0
     + IORQ * RD * A7 * !A6 * !A5 * A4 * !A3 * A2 * !S2 * !S1 * !S0;

```

```
'Reseterzeugung:
```

```
RESET = PCL;
```

```
%END
```

.. Bauteileliste ADI-12 -----  
 |  
 -----

1*		Platine ADI-12
1*	ST1	Messerleiste DIN 41.612 Bauform C, 64polig, gewinkelt
1*	ST4	D-Sub-Stiftleiste, 9polig (DB9), gewinkelte Lötstifte
1*	ST5	D-Sub-Buchsenleiste, 15polig (DB15), gewinkelte Lötstifte
1*		Kunststoffwinkel für Frontplatte
2*		Zylinderkopfschraube M3*8
2*		Mutter sechskant M3
2*		Zylinderkopfschraube M2,5*10
2*		Zylinderkopfschraube M2,5*8
4*		Mutter sechskant M2,5
1*		Stiftleiste gerade, 2,54mm, zweireihig, 58polig
1*	ST2	Stiftleiste oder Pfostenwanne, 10polig, ohne Auswerfer
1*	ST3	Stiftleiste oder Pfostenwanne, 26polig, ohne Auswerfer
16*		Jumper
2*	RN1, RN4	Widerstandsarray SIL 4* 10k
2*	RN2, RN3	Widerstandsarray SIL 8* 10k
1*		Widerstand Metallfilm 0R (15k)
9*		Widerstand Metallfilm 100R
8*		Widerstand Metallfilm 1k2
8*		Widerstand Metallfilm 10k
8*		Widerstand Metallfilm 22k
8*		Widerstand Metallfilm 100k
1*	RINT	Widerstand Metallfilm 200k (20k/100k)
1*		Widerstand Metallfilm 1M
1*		Widerstand Metallfilm 10M
1*		Spindeltrimmer 10k stehend, Zehngang
19*		Keramikkondensator u1 RM 2,54 (Z5U o.ä.)
1*	CINT	Folienkondensator MKS 220n (150n) RM 5..10 (besser MKP!?)
1*	CAZ	Folienkondensator MKS 470n (330n) RM 5..10 (auch MKT)
1*		Folienkondensator MKS 1u RM 5..10 (auch MKT)
3*		Elko 10u RM 2,54
1*		Elko 100u RM 2,54
1*		IC 74 LS 245 (oder HC/ HCT)
1*		IC GAL 16V8 (programmiert f. ADI-12)
1*		IC 8255 (NEC 71055)
1*		IC 7660
1*		IC ICL 7109
1*		IC MX 580KH (auch AD-, LT- 580 oder PMI)
1*		IC DG 508 (oder ADG 508, HI 508, MAX358)
2*		Vierfach Optokoppler PC 847
8*		Transistor BC 880
1*		Quarz 3MHz HC-18/U
1*		Präz. socket 8pol.
3*		Präz. socket 16pol.
2*		Präz. socket 20pol.
2*		Präz. socket 40pol.

.- **Literaturliste** -----  
|  
-----

- [1] n.n.: NEC Microprocessors and Peripherals Data Book. NEC ohne Jahresang.  
uPD71055 Seite 6.49
- [2] n.n.: MAXIM 1990 Integrated Circuits Data Book. MAXIM 1990  
Seite 1-131, 3-11, 11-27
- [3] Tietze U.; Schenk Ch.: Halbleiter-Schaltungstechnik. Berlin,..: Springer-  
Verlag 1990, ISBN: 3-540-19475-4
- [4] Bergman K.: Elektrische Meßtechnik. Braunschweig: Vieweg-Verlag 1986,  
ISBN: 3-528-24080-6

Die Kopien des Datenblattes "ICL 7109" stammen aus folgender Quelle:

- [5] n.n.: INTERSIL Data Book. INTERSIL; Spezial-Elektronik 1982